
Jの仕事初めに曜日とカレンダーの印刷
Zellerの公式による計算

西川　利男

　正月とて、近くの神社に初詣に行き、ベンチで日向ぼっこしながら、スマホでJを

いじっている。今年の初仕事として、曜日からカレンダーを作ってみた。

　数学の楽しみは、自分で計算をやってみなくてはわからない。しかし、電卓ではた

いしたことはできないし、パソコンでBASICや Cのプログラムはおっくうである。

　スマホの上でも可能なJは最適な環境である。楽しむのであるから、易しいところ

から始めたらよい。

１．君の誕生日は何曜日？－小学生の算数の問題として

　今年、2017年 1月 1日は日曜日から始まる。したがって、曜日を知るには、元日か

ら何日目になるかを知って、7で割った余りを求めれば良い。

　つまり7を法としたmodの計算が必要であり、Jでは極めて簡単に行える。

まず、1月の最初の日でテストしてみる。

　　7|1 => 1(日), 7|2 => 2(月), .., 7|7 => 0(土) , 7|8 => 1(日)

　さて、私の誕生日4月27日を例に、Jでやってみよう。

　　M =: 4

    D =: 27

    MD =: 31 28 31 30 31 30 31 31 30 31 30 31

    DD =: D + +/(i. <: M) { MD

    DD => 117 (元日から117日)

  従って、

    7|DD => 5(木)　　　　

となる。    

　ついでに4月のカレンダーは次のようになる。（ブランクは_(数値∞)で示した。）

   (6, 7) $ (6#_), (>: i. <: M { MD), (7#_)  

 _  _  _  _  _  _  1

 2  3  4  5  6  7  8

 9 10 11 12 13 14 15

16 17 18 19 20 21 22

23 24 25 26 27 28 29

30  _  _  _  _  _  _

この後、年月日から曜日を求めるZellerの公式を用いて、体裁良くふつうの形で、

曜日とカレンダーを印刷する処理をJでやってみよう。

２．Zellerの公式

　西暦の年月日から曜日を算出する公式として、Zellerの公式が知られている。

　y年m月d日の曜日を求めるZellerの公式は次のようになる。ただし、１月と２月

とは、前の年の13月と14月として計算する。たとえば、2017年 1月 1日は2016年 13

月 1日に、2017年 2月 1日は2016年 14月 1日のようになる。

また

  　 金木水火月日土

6543210h

に対応する。

３．Jによるゼラーの公式をもとに年月日から曜日の計算

zellh =: 3 : 0

'y m d' =. y.

C =. <. y % 100

Y =. 100 | y

h =. (5*C) + Y + (<. Y % 4) + (<. C % 4)

h =. h + <. 2.6*(m+1)

h =. h + d

7 | h 

)

week =: 3 : 0

'y m d' =. y.

if. m < 3 do.

  y =. y - 1

  m =. m + 12

end.

w =. zellh y, m, d

W =: 7 | 6 + w

> W { 'Sun';'Mon';'Tue';'Wed';'Thu';'Fri';'Sat'

)

　次のように実行される。

  week 2017, 1, 1 

Sun 

  week 2017, 4, 27

Thu

４．Jによる曜日からカレンダーの計算と印刷

　カレンダーの処理とは、計算というより文字列の並びを配列として、どう表の形に

整えて印刷するかというJプログラミングである。

　例えば、小の月(30)や大の月(31)に加えて、28や 29の月もあり、また1日が週の初

めか、終わりかでカレンダー配列の行数が変化するので、体裁を整えるのは意外とや

っかいである。

SP10 =: 10#' '

MON =: (SP10,'January');(SP10,'February');(SP10,'March')

MON =: MON, (SP10,'April');(SP10,'May');(SP10,'June') 

MON =: MON, (SP10,'July');(SP10,'August');(SP10,'September')

MON =: MON, (SP10,'October');(SP10,'November');(SP10,'December')

calendar =: 3 : 0

'y m' =. y.

week y, m, 1

MM =. (m-1){ MD

NB. if. w = 0 do. w =. 7 end.

CA =: (6, 7) $ ((W)#_), (>: i.  MM), (14#_)

if. */ (({: CA) e. _) do. CA =: }: CA end.

CAL =. ' Sun Mon Tue Wed Thu Fri Sat' , 4j0 ": CA

}."(1) (> (<: m) {MON), CAL

) 

実行すると、次のようになる。

   calendar 2017, 1

         January           

Sun Mon Tue Wed Thu Fri Sat

  1   2   3   4   5   6   7

  8   9  10  11  12  13  14

 15  16  17  18  19  20  21

 22  23  24  25  26  27  28

 29  30  31   _   _   _   _

　これを使って、一年分のカレンダーを印刷するプログラムを作った。

calendlist =: 3 : 0

year =. y.

CALEN =. ''

m =. 1

while. m < 13

  do.

    CA =. calendar (year, m)

    CALEN =. CALEN, < CA

    m =. m + 1

  end.

(6, 2) $ CALEN

)

　以下に2017年の一年分のカレンダーを表示した。

   calendlist 2017
+---------------------------+---------------------------+

|         January           |         February          |

|Sun Mon Tue Wed Thu Fri Sat|Sun Mon Tue Wed Thu Fri Sat|

|  1   2   3   4   5   6   7|  _   _   _   1   2   3   4|

|  8   9  10  11  12  13  14|  5   6   7   8   9  10  11|

| 15  16  17  18  19  20  21| 12  13  14  15  16  17  18|

| 22  23  24  25  26  27  28| 19  20  21  22  23  24  25|

| 29  30  31   _   _   _   _| 26  27  28   _   _   _   _|

+---------------------------+---------------------------+

|         March             |         April             |

|Sun Mon Tue Wed Thu Fri Sat|Sun Mon Tue Wed Thu Fri Sat|

|  _   _   _   1   2   3   4|  _   _   _   _   _   _   1|

|  5   6   7   8   9  10  11|  2   3   4   5   6   7   8|

| 12  13  14  15  16  17  18|  9  10  11  12  13  14  15|

| 19  20  21  22  23  24  25| 16  17  18  19  20  21  22|

| 26  27  28  29  30  31   _| 23  24  25  26  27  28  29|

|                           | 30   _   _   _   _   _   _|

+---------------------------+---------------------------+

|         May               |         June              |

|Sun Mon Tue Wed Thu Fri Sat|Sun Mon Tue Wed Thu Fri Sat|

|  _   1   2   3   4   5   6|  _   _   _   _   1   2   3|

|  7   8   9  10  11  12  13|  4   5   6   7   8   9  10|

| 14  15  16  17  18  19  20| 11  12  13  14  15  16  17|

| 21  22  23  24  25  26  27| 18  19  20  21  22  23  24|

| 28  29  30  31   _   _   _| 25  26  27  28  29  30   _|

+---------------------------+---------------------------+

|         July              |         August            |

|Sun Mon Tue Wed Thu Fri Sat|Sun Mon Tue Wed Thu Fri Sat|

|  _   _   _   _   _   _   1|  _   _   1   2   3   4   5|

|  2   3   4   5   6   7   8|  6   7   8   9  10  11  12|

|  9  10  11  12  13  14  15| 13  14  15  16  17  18  19|

| 16  17  18  19  20  21  22| 20  21  22  23  24  25  26|

| 23  24  25  26  27  28  29| 27  28  29  30  31   _   _|

| 30  31   _   _   _   _   _|                           |

+---------------------------+---------------------------+

|         September         |         October           |

|Sun Mon Tue Wed Thu Fri Sat|Sun Mon Tue Wed Thu Fri Sat|

|  _   _   _   _   _   1   2|  1   2   3   4   5   6   7|

|  3   4   5   6   7   8   9|  8   9  10  11  12  13  14|

| 10  11  12  13  14  15  16| 15  16  17  18  19  20  21|

| 17  18  19  20  21  22  23| 22  23  24  25  26  27  28|

| 24  25  26  27  28  29  30| 29  30  31   _   _   _   _|

+---------------------------+---------------------------+

|         November          |         December          |

|Sun Mon Tue Wed Thu Fri Sat|Sun Mon Tue Wed Thu Fri Sat|

|  _   _   _   1   2   3   4|  _   _   _   _   _   1   2|

|  5   6   7   8   9  10  11|  3   4   5   6   7   8   9|

| 12  13  14  15  16  17  18| 10  11  12  13  14  15  16|

| 19  20  21  22  23  24  25| 17  18  19  20  21  22  23|

| 26  27  28  29  30   _   _| 24  25  26  27  28  29  30|

|                           | 31   _   _   _   _   _   _|

+---------------------------+---------------------------+   

NB. Calendar and Week         

NB. Zeller's Congruence

NB.   by T.N 2016/12/28

M =: 4

D =: 27

MD =: 31 28 31 30 31 30 31 31 30 31 30 31

DD =: D + +/(i. <: M) { MD

Cal4 =:(6, 7) $ (6#_), (>: i. <: M { MD), (7#_) 

wr =: 1!:2&2

NB. Zeller's Congruence　= Calculate Week Day

NB.  zellh 2016 12 28 => 4 .. 水曜日

NB.  zellh 2016 14 1 =>  1 .. 日曜日

NB.  zellh 2017 4 27 =>  5 .. 木曜日

zellh =: 3 : 0

'y m d' =. y.

C =. <. y % 100

Y =. 100 | y

h =. (5*C) + Y + (<. Y % 4) + (<. C % 4)

h =. h + <. 2.6*(m+1)

h =. h + d

7 | h 

)

NB.  week 2017, 1, 1 => Sun 

week =: 3 : 0

'y m d' =. y.

if. m < 3 do.

  y =. y - 1

  m =. m + 12

end.

w =. zellh y, m, d

W =: 7 | 6 + w

> W { 'Sun';'Mon';'Tue';'Wed';'Thu';'Fri';'Sat'

)

SP10 =: 10#' '

MON =: (SP10,'January');(SP10,'February');(SP10,'March')

MON =: MON, (SP10,'April');(SP10,'May');(SP10,'June') 

MON =: MON, (SP10,'July');(SP10,'August');(SP10,'September')

MON =: MON, (SP10,'October');(SP10,'November');(SP10,'December')

calendar =: 3 : 0

'y m' =. y.

week y, m, 1

MM =. (m-1){ MD

NB. if. w = 0 do. w =. 7 end.

CA =: (6, 7) $ ((W)#_), (>: i.  MM), (14#_)

if. */ (({: CA) e. _) do. CA =: }: CA end.

CAL =. ' Sun Mon Tue Wed Thu Fri Sat' , 4j0 ": CA

}."(1) (> (<: m) {MON), CAL

) 

calendlist =: 3 : 0

year =. y.

NB. JAN =. calendar (year, 1)

NB. FEB =. calendar (year, 2)

NB. MAR =. calendar (year, 3)

NB. JAN;FEB;MAR

CALEN =. ''

m =. 1

while. m < 13

  do.

    CA =. calendar (year, m)

    CALEN =. CALEN, < CA

    m =. m + 1

  end.

(6, 2) $ CALEN

)

